
//Hexapod Slave 105 (Arduino uno)

#include <Servo.h>
#include <Wire.h>

// Definiamo gli oggetti che controllano i servomotori
Servo myservo[3][3]; 

// Definiamo i piedini ai quali sono collegati i servomotori 
byte myservopins[3][3]={2,3,4,5,6,7,8,9,10};

byte cont=0;

// Posizione attuale
byte pos[3][3];

// Posizione da raggiungere
int goToPos[3][3]; 

// Posizione di default dei servomotori
const byte defaultPosition[3][3]={90,60,120,90,60,120,90,60,120};

void setup() {
  byte i,j;

  // Piedini tasti come ingressi
  pinMode(14,INPUT_PULLUP);
  pinMode(15,INPUT_PULLUP);
  
  // Imposto l’indirizzo dello slave
  Wire.begin(0x04);

  // Definisci velocità di trasmissione
  Serial.begin(9600);
  
  // Evento per la ricezione del dato
  Wire.onReceive(receiveEvent);
  
  // Evento per la richiesta del dato  
  Wire.onRequest(requestEvent);

  // Indichiamo i piedini ai quali gli oggetti che controllano i servomotori fanno riferimento
  for(i=0;i<3;i++) for(j=0;j<3;j++) myservo[i][j].attach(myservopins[i][j]);  
  
  // Carichiamo la posizione iniziale
  for(i=0;i<3;i++) for(j=0;j<3;j++) goToPos[i][j]=defaultPosition[i][j];

  delay(1000);
}

// Evento viene generato quando il master richiede un dato
void requestEvent(){
  //spedisco il dato al Master
  Wire.write(pos[cont/3][cont%3]);
}

// Evento generato quando sul bus è presente un dato da leggere
void receiveEvent(int data){
  byte x;
  //eseguo la lettura
  x = Wire.read();
  cont=x%10;
  x=x-cont;
  goToPos[cont/3][cont%3]=x;
}

// Raggiungi la posizione per i servomotori della scheda slave
void raggiungiPosizione(){
  byte i,j;
  for(i=0;i<3;i++) for(j=0;j<3;j++) 
  {
    pos[i][j]=myservo[i][j].read();
    if(pos[i][j]!=goToPos[i][j]) myservo[i][j].write(goToPos[i][j]);
  }



}

void loop() { 
  byte i,j;
  
  // Raggiungi la posizione
  raggiungiPosizione();
  
  delay(20);
}


